# Maintainer: Morten Linderud <foxboron@archlinux.org>

pkgname=python-milc
pkgver=1.6.6
pkgrel=2
pkgdesc="Opinionated Batteries-Included Python 3 CLI Framework."
arch=("any")
url="https://milc.clueboard.co/#/"
license=("MIT")
depends=("python-appdirs" "python-argcomplete" "python-colorama" "python-halo")
makedepends=("python-setuptools")
checkdepends=("python-semver" "python-nose2")
source=("$pkgname-$pkgver.tar.gz::https://github.com/clueboard/milc/archive/${pkgver}.tar.gz")
sha256sums=('b383bccac8dfb21748ec0433daba2c9128b53c2991b5c443284d102ad219841b')

build() {
    cd "${pkgname#python-}-${pkgver}"
    python setup.py build
}

check() {
    cd "${pkgname#python-}-${pkgver}"
    rm tests/test_script*.py
    nose2
}

package() {
    cd "${pkgname#python-}-${pkgver}"
    python setup.py install --root="${pkgdir}/" --optimize=1 --skip-build
    install -Dm644 LICENSE "${pkgdir}/usr/share/licenses/${pkgname}/LICENSE"
}

# vim:set ts=4 sw=4 et:
